

mepoisson postestimation — Postestimation tools for mepoisson

[Postestimation commands](#)
[predict](#)
[margins](#)
[Remarks and examples](#)
[Methods and formulas](#)
[Also see](#)

Postestimation commands

The following postestimation command is of special interest after `mepoisson`:

| Command | Description |
|--------------------------|---|
| <code>estat group</code> | summarize the composition of the nested groups |
| <code>estat sd</code> | display variance components as standard deviations and correlations |

The following standard postestimation commands are also available:

| Command | Description |
|------------------------------|---|
| <code>contrast</code> | contrasts and ANOVA-style joint tests of estimates |
| <code>estat ic</code> | Akaike's and Schwarz's Bayesian information criteria (AIC and BIC) |
| <code>estat summarize</code> | summary statistics for the estimation sample |
| <code>estat vce</code> | variance–covariance matrix of the estimators (VCE) |
| <code>estat (svy)</code> | postestimation statistics for survey data |
| <code>estimates</code> | cataloging estimation results |
| * <code>hausman</code> | Hausman's specification test |
| <code>lincom</code> | point estimates, standard errors, testing, and inference for linear combinations of coefficients |
| * <code>lrtest</code> | likelihood-ratio test |
| <code>margins</code> | marginal means, predictive margins, marginal effects, and average marginal effects |
| <code>marginsplot</code> | graph the results from margins (profile plots, interaction plots, etc.) |
| <code>nlcom</code> | point estimates, standard errors, testing, and inference for nonlinear combinations of coefficients |
| <code>predict</code> | predictions, residuals, influence statistics, and other diagnostic measures |
| <code>predictnl</code> | point estimates, standard errors, testing, and inference for generalized predictions |
| <code>pwcompare</code> | pairwise comparisons of estimates |
| <code>test</code> | Wald tests of simple and composite linear hypotheses |
| <code>testnl</code> | Wald tests of nonlinear hypotheses |

* `hausman` and `lrtest` are not appropriate with `svy` estimation results.

predict

Description for predict

`predict` creates a new variable containing predictions such as mean responses; linear predictions; density and distribution functions; standard errors; and Pearson, deviance, and Anscombe residuals.

Menu for predict

Statistics > Postestimation

Syntax for predict

Syntax for obtaining predictions of the outcome and other statistics

```
predict [type] newvarsspec [if] [in] [, statistic options]
```

Syntax for obtaining estimated random effects and their standard errors

```
predict [type] newvarsspec [if] [in], reffects [re_options]
```

Syntax for obtaining ML scores

```
predict [type] newvarsspec [if] [in], scores
```

newvarsspec is *stub** or *newvarlist*.

| <i>statistic</i> | Description |
|------------------|-------------|
|------------------|-------------|

| | |
|---------------------------|--|
| Main | |
| <code>mu</code> | mean response; the default |
| <code>eta</code> | fitted linear predictor |
| <code>xb</code> | linear predictor for the fixed portion of the model only |
| <code>stdp</code> | standard error of the fixed-portion linear prediction |
| <code>density</code> | predicted density function |
| <code>distribution</code> | predicted distribution function |
| <code>pearson</code> | Pearson residuals |
| <code>deviance</code> | deviance residuals |
| <code>anscombe</code> | Anscombe residuals |

These statistics are available both in and out of sample; type `predict ... if e(sample) ...` if wanted only for the estimation sample.

| <i>options</i> | Description |
|--|--|
| Main | |
| <code>conditional(<i>ctype</i>)</code> | compute <i>statistic</i> conditional on estimated random effects; default is <code>conditional(ebmeans)</code> |
| <code>marginal</code> | compute <i>statistic</i> marginally with respect to the random effects |
| <code>nooffset</code> | make calculation ignoring offset or exposure |
| Integration | |
| <code>int_options</code> | integration options |
| pearson, deviance, anscombe may not be combined with marginal. | |
| <i>ctype</i> | Description |
| <code>ebmeans</code> | empirical Bayes means of random effects; the default |
| <code>ebmodes</code> | empirical Bayes modes of random effects |
| <code>fixedonly</code> | prediction for the fixed portion of the model only |
| <i>re_options</i> | Description |
| Main | |
| <code>ebmeans</code> | use empirical Bayes means of random effects; the default |
| <code>ebmodes</code> | use empirical Bayes modes of random effects |
| <code>reses(<i>stub*</i> <i>newvarlist</i>)</code> | calculate standard errors of empirical Bayes estimates |
| Integration | |
| <code>int_options</code> | integration options |
| <i>int_options</i> | Description |
| <code>intpoints(#)</code> | use # quadrature points to compute marginal predictions and empirical Bayes means |
| <code>iterate(#)</code> | set maximum number of iterations in computing statistics involving empirical Bayes estimators |
| <code>tolerance(#)</code> | set convergence tolerance for computing statistics involving empirical Bayes estimators |

Options for predict

Main

`mu`, the default, calculates the predicted mean, that is, the predicted number of events.

`eta`, `xb`, `stdp`, `density`, `distribution`, `pearson`, `deviance`, `anscombe`, `scores`, `conditional()`, `marginal`, and `nooffset`; see [ME] [meglm postestimation](#).

`reffects`, `ebmeans`, `ebmodes`, and `reses()`; see [ME] [meglm postestimation](#).

Integration

`intpoints()`, `iterate()`, and `tolerance()`; see [ME] [meglm postestimation](#).

margins

Description for margins

`margins` estimates margins of response for mean responses and linear predictions.

Menu for margins

Statistics > Postestimation

Syntax for margins

```
margins [marginlist] [, options]
```

```
margins [marginlist] , predict(statistic ...) [predict(statistic ...) ...] [options]
```

| <i>statistic</i> | Description |
|---------------------------|--|
| <code>mu</code> | mean response; the default |
| <code>eta</code> | fitted linear predictor |
| <code>xb</code> | linear predictor for the fixed portion of the model only |
| <code>stdp</code> | not allowed with <code>margins</code> |
| <code>density</code> | not allowed with <code>margins</code> |
| <code>distribution</code> | not allowed with <code>margins</code> |
| <code>pearson</code> | not allowed with <code>margins</code> |
| <code>deviance</code> | not allowed with <code>margins</code> |
| <code>anscombe</code> | not allowed with <code>margins</code> |
| <code>reflects</code> | not allowed with <code>margins</code> |
| <code>scores</code> | not allowed with <code>margins</code> |

Options `conditional(ebmeans)` and `conditional(ebmodes)` are not allowed with `margins`.

Option `marginal` is assumed where applicable if `conditional(fixedonly)` is not specified.

Statistics not allowed with `margins` are functions of stochastic quantities other than $e(b)$.

For the full syntax, see [\[R\] margins](#).

Remarks and examples

[stata.com](#)

Various predictions, statistics, and diagnostic measures are available after fitting a mixed-effects Poisson model with `mepoisson`. For the most part, calculation centers around obtaining estimates of the subject/group-specific random effects. Random effects are not estimated when the model is fit but instead need to be predicted after estimation.

Here we show a short example of predicted counts and predicted random effects; refer to [\[ME\] meglm postestimation](#) for additional examples applicable to mixed-effects generalized linear models.

▶ Example 1: Predicting counts and random effects

In example 2 of [ME] `mepoisson`, we modeled the number of observed epileptic seizures as a function of treatment with the drug progabide and other covariates,

$$\log(\mu_{ij}) = \beta_0 + \beta_1 \text{treat}_{ij} + \beta_2 \text{lbas}_{ij} + \beta_3 \text{lbas_trt}_{ij} + \beta_4 \text{lage}_{ij} + \beta_5 \text{visit}_{ij} + u_j + v_j \text{visit}_{ij}$$

where (u_j, v_j) are bivariate normal with 0 mean and variance–covariance matrix

$$\Sigma = \text{Var} \begin{bmatrix} u_j \\ v_j \end{bmatrix} = \begin{bmatrix} \sigma_u^2 & \sigma_{uv} \\ \sigma_{uv} & \sigma_v^2 \end{bmatrix}$$

```
. use http://www.stata-press.com/data/r15/epilepsy
(Epilepsy data; progabide drug treatment)
. mepoisson seizures treat lbas lbas_trt lage visit || subject: visit,
> cov(unstructured) intpoints(9)
(iteration log omitted)

Mixed-effects Poisson regression          Number of obs    =          236
Group variable:          subject          Number of groups =           59
Obs per group:
      min =          4
      avg =         4.0
      max =          4

Integration method: mvaghermite          Integration pts. =           9
Wald chi2(5)              =        115.56
Prob > chi2                =         0.0000

Log likelihood = -655.68103
```

| seizures | Coef. | Std. Err. | z | P> z | [95% Conf. Interval] | |
|-------------------|-----------|-----------|-------|-------|----------------------|-----------|
| treat | -.9286592 | .4021715 | -2.31 | 0.021 | -1.716901 | -.1404175 |
| lbas | .8849762 | .1312535 | 6.74 | 0.000 | .627724 | 1.142228 |
| lbas_trt | .3379759 | .2044471 | 1.65 | 0.098 | -.062733 | .7386849 |
| lage | .4767192 | .3536276 | 1.35 | 0.178 | -.2163781 | 1.169817 |
| visit | -.2664098 | .1647098 | -1.62 | 0.106 | -.5892352 | .0564156 |
| _cons | 2.099555 | .2203749 | 9.53 | 0.000 | 1.667629 | 2.531482 |
| subject | | | | | | |
| var(visit) | .5314803 | .229385 | | | .2280928 | 1.238405 |
| var(_cons) | .2514923 | .0587902 | | | .1590534 | .3976549 |
| subject | | | | | | |
| cov(visit, _cons) | .0028715 | .0887037 | 0.03 | 0.974 | -.1709846 | .1767276 |

```
LR test vs. Poisson model: chi2(3) = 324.54          Prob > chi2 = 0.0000
```

Note: LR test is conservative and provided only for reference.

The purpose of this model was to allow subject-specific linear log trends over each subject's four doctor visits, after adjusting for the other covariates. The intercepts of these lines are distributed $N(\beta_0, \sigma_u^2)$, and the slopes are distributed $N(\beta_5, \sigma_v^2)$, based on the fixed effects and assumed distribution of the random effects.

We can use `predict` to obtain estimates of the random effects u_j and v_j and combine these with our estimates of β_0 and β_5 to obtain the intercepts and slopes of the linear log trends.

```

. predict re_visit re_cons, reffects
(calculating posterior means of random effects)
(using 9 quadrature points)
. generate b1 = _b[visit] + re_visit
. generate b0 = _b[_cons] + re_cons
. by subject, sort: generate tolist = _n==1
. list subject treat b1 b0 if tolist & (subject <=5 | subject >=55)

```

| | subject | treat | b1 | b0 |
|------|---------|-------|-----------|----------|
| 1. | 1 | 0 | -.428854 | 2.13539 |
| 5. | 2 | 0 | -.2731013 | 2.149744 |
| 9. | 3 | 0 | .0022089 | 2.417506 |
| 13. | 4 | 0 | -.3197094 | 2.238224 |
| 17. | 5 | 0 | .6082718 | 2.110739 |
| 217. | 55 | 1 | -.2308834 | 2.282539 |
| 221. | 56 | 1 | .2912798 | 3.19678 |
| 225. | 57 | 1 | -.4828764 | 1.423153 |
| 229. | 58 | 1 | -.2519466 | 1.131373 |
| 233. | 59 | 1 | -.1269573 | 2.171541 |

We list these slopes (b1) and intercepts (b0) for five control subjects and five subjects on the treatment.

```

. count if tolist & treat
31
. count if tolist & treat & b1 < 0
25
. count if tolist & !treat
28
. count if tolist & !treat & b1 < 0
20

```

We also find that 25 of the 31 subjects taking progabide were estimated to have a downward trend in seizures over their four doctor visits, compared with 20 of the 28 control subjects.

We also obtain predictions for number of seizures, and unless we specify the `conditional(fixedonly)` option, these predictions will incorporate the estimated subject-specific random effects.

```

. predict n
(predictions based on fixed effects and posterior means of random effects)
(option mu assumed)
(using 9 quadrature points)

```

```
. list subject treat visit seizures n if subject <= 2 | subject >= 58, sep(0)
```

| | subject | treat | visit | seizures | n |
|------|---------|-------|-------|----------|----------|
| 1. | 1 | 0 | -.3 | 5 | 3.775774 |
| 2. | 1 | 0 | -.1 | 3 | 3.465422 |
| 3. | 1 | 0 | .1 | 3 | 3.18058 |
| 4. | 1 | 0 | .3 | 3 | 2.919151 |
| 5. | 2 | 0 | -.3 | 3 | 3.598805 |
| 6. | 2 | 0 | -.1 | 5 | 3.40751 |
| 7. | 2 | 0 | .1 | 3 | 3.226382 |
| 8. | 2 | 0 | .3 | 3 | 3.054883 |
| 229. | 58 | 1 | -.3 | 0 | .9611137 |
| 230. | 58 | 1 | -.1 | 0 | .9138838 |
| 231. | 58 | 1 | .1 | 0 | .8689747 |
| 232. | 58 | 1 | .3 | 0 | .8262726 |
| 233. | 59 | 1 | -.3 | 1 | 2.40652 |
| 234. | 59 | 1 | -.1 | 4 | 2.346184 |
| 235. | 59 | 1 | .1 | 3 | 2.287361 |
| 236. | 59 | 1 | .3 | 2 | 2.230013 |

◀

Methods and formulas

Methods and formulas for predicting random effects and other statistics are given in *Methods and formulas* of [ME] [meglm postestimation](#).

Also see

[ME] [mepoisson](#) — Multilevel mixed-effects Poisson regression

[ME] [meglm postestimation](#) — Postestimation tools for meglm

[U] [20 Estimation and postestimation commands](#)